LYNX DISK USER GROUP NEWSLETTER No 1

43 Church Road

Cowley

Uxbridge

Middx

UB8 3ND

                 6th August 1993

Dear Lynx User,

It is now over a year since the LDUG Disk no 2 was released.

Not wanting to lose the enthusiasm of the small band of committed Lynx Users (ever decreasing!) who contributed to the LRG\LDUG volumes, I decided to attempt to keep your interest in the Lynx, by producing a newsletter of this type.

Hopefully, the newsletter can be a source of news, tips and information all relating to the Camputers Lynx. It is feedback from you that determines whether this newsletter fails or succeeds.

It is intended that the newsletter will be accompanied by a disk in future, a disk that is supplied by you. All you need to do is send your contributions on disk (with the return postage) to the address at the top of the page. I will then compile the next volume in much the same way as John Koushappas did in the past, but with the following differences.

    1. All document/text files will be printed out on cheap listing


  paper like this. This should mean that it will no longer be


  necessary to include READ.TEXT or LTF files on the disk.


  Good riddance! If it is found that editorial control is needed,


  then so be it. BASIC programs, Source\Object files etc will


  continue to be included on the disk.

    2. The disk will be loosely compiled, that is there will be no


  formally produced menus or glitzy screen shots to go through.

    3. If a submitted program\routine requires the need of a hardware


  modification to be present in order to work successfully, then


  this fact will be highlighted somewhere in the contents of the


  newsletter. Gone are the days whereby, users found that they


  could not explain why program XYZ would not work on their


  machines.

I feel that is all I can say about what the appearance of the newsletter will look like in future. If you have any comments or suggestions about this please do not hesitate to contact me.

I hope somewhere in the pages of this first newsletter, you'll find something of interest.







Regards, Martyn Smith

This newsletter was produced using WordPerfect 5.1 on a PC. All Lynx listings present were transferred to the PC in their original form using a PC file transfer program called PROCOMM.

NEWS

The only piece of news I have at present is sad news. Colin Clayman (of Scorpion Rom and CP/M on the 96K fame) has recently sold all his Lynx equipment.

Colin, as you may know was very knowledgable on aspects of programming on the Lynx, not least Lynx DOS. Without his paper on Lynx DOS, I for one would have never ventured into writing routines that made use of Lynx DOS.

TIPS

1. SPECCY.TRANSFER

This program was included on Disk No 3 of LRG. It allows Sinclair Spectrum tapes to be readable on the 96K Lynx. In fact, I used it to transfer ESPIONAGE ISLAND (LRG Disk No 4). If you have attempted to use it yourself, you may have noticed that it is not possible to load the very first program on a tape. This two line patch will correct this.


105 IF b=0 THEN POKE &1034,&1B


106 ELSE POKE &1034,&0E

2. EXT COPY and EXT SCOPY from the Utilities Disk (ver 0.02)

Have you ever got the error message "DOS Error &93 File not found"

when the file on the source disk does clearly exist? I only discovered this bug when I recently started to make backup copies of LRG\LDUG disks! It appears that EXT COPY and EXT SCOPY have trouble with filenames that have more than 26 characters in them. Unfortunately, all my attempts to correct this problem have been fruitless. Therefore the only tip I have to offer on this is, keep your filenames short(er) 

3. Displaying text from machine code (96K\128K compatible)

A handy little routine resides in the high ram DOS routines. It is located at &FD13. To use the routine, prime HL with a pointer to the text (the text should be terminated with a null) and call &FD13 ie


LD
HL,text1


CALL &FD13


RET

text1:
DB "Some text to display",&0D,0

An added bonus is also included - the same routine can be used to display (in hex) the values of the registers DE and A at the same time as displaying "normal" text ie


LD
DE,1967


LD
A,26


LD
HL,text2


CALL &FD13


RET

text2:
DB "I was born in ",&0B,", so that makes me ",3,&0D,0

BOOT MAKER






96K\128K Lynx compatible

One of the most underused commands in Lynx DOS must be EXT BOOT. Unless you have CP/M, I'm sure each time you type EXT BOOT <cr>, you probably get an error message "DOS Error &B5 Disk type" ?

This error message is a bit misleading. It does not mean that DOS disks cannot be BOOTed, it simply means that there is no boot code on the disk! With the accompanying utility, &B5 errors may well be a thing from the past.

What does BOOTMAKER do? Well it patches your DOS disks with a short routine that loads a BASIC program called "BOOT" from the default disk each time EXT BOOT <cr> is typed. If the BASIC program "BOOT" is not found then, an error message "DOS Error &93 File not found" will be given and control will be regained by BASIC. If the BASIC program "BOOT" is found then, it will be loaded and executed (if saved with auto run line number !)

The contents of the BASIC program "BOOT" can be any combination of BASIC commands. An example is given on the following pages.


Instructions

1. Type in the BOOTMAKER.BAS It is essential that the program is

   entered as supplied here - especially the CODE lines.

2. Save BOOTMAKER.BAS to disk before running it. Safety precaution !

3. Execute BOOTMAKER.BAS with RUN.

4. Those of you with a nervous disposition should practice on a

   scratch disk first.

5. Follow the instructions as given in BOOTMAKER.BAS

6. Create your BASIC program "BOOT" as required. Save it on the same

   disk you have just patched !!!


Notes on using EXT BOOT

1. The only methods EXT BOOT can be entered from the keyboard as, are


a. EXT BOOT 
( abbreviation EXT BO. )


b. EXT DRIVE "B:"


   EXT BOOT

2. EXT DRIVE "A:"

   EXT BOOT "B:" 

   - will only partially work; what will actually occur is, drive B:

   will be BOOTed but the search for the BASIC program "BOOT" will be

   on drive A:

Explanation of BOOTMAKER

To understand how the utility works it is first necessary to explain what EXT BOOT actually does and how block 0 on every Lynx DOS disk is organised.


Pseudo code for the EXT BOOT command

1. Open selected disk for reading.

2. Load blocks 0 and 1 from disk into memory (Address &4633 or &4833)

3. If byte at offset 2 (starting from 0) is equal to &18 (JR opcode)

   then execute the jump relative instruction

4. If byte at offset 2 is not equal to &18, then return &B5 error

5. Close disk

6. Return to caller


Structure of block 0

Bytes 00 - 01

Disk format (0 = CP/M, 1 = DOS)

Bytes 02 - 03

Boot instruction 

Byte  04


Not used

Bytes 05 - 07

Disk type (81,00,1B = 200K / 8C,40,40 = 800K)

Bytes 08 - 0B

Block no. of Index (word reversed, byte reversed)

Bytes 0C - 0F

No. blocks on disk (word reversed, byte reversed)

Bytes 10 - 13

Disk number

Bytes 14 - 3D

Copyright text

Bytes 3E - FF

Available for boot code


(Reproduced from "Programming for Lynx DOS disks" by Colin Clayman)


Pseudo code for BOOTMAKER

1. Reset disk in drive (this ensures erroneous data from a previous






disk isn't used)

2. Load blocks 0 and 1 into memory at &8000

3. Ensure disk is a DOS disk by checking the bytes at &8000+0 and

   &8000+1

4. Put the bytes &18 and &3B at &8000+2 and &8000+3 ie JR &3B

5. Copy the boot loader and boot execution code (located in line 120)

   to &8000+&3E

6. Write the modified blocks 0 and 1 back to disk


Pseudo code for the boot loader

1. Calculate location of where the boot loader itself resides

2. Copy the boot execution code into visible memory

3. Put the address of the boot execution code on the users stack

4. Perform a normal exit from EXT BOOT


Pseudo code for boot execution

5. Close the previously opened disk

6. Simulate EXT LOAD "BOOT"

1   REM BOOTMAKER.BAS

100 CODE  CD 04 40 EB 01 00 00 CD F2 FF 00 6F AF 67 C3 0E 40

110 CODE  00 00 00 00 00

120 CODE  21 00 00 44 4D 3E B7 ED B1 2B 36 00 B7 77 23 20 F6 11 1B 00 
19 11 D8 FE 01 10 00 ED B0 2A E6 40 11 D8 FE 73 23 72 AF C9 CD 78 
FC 11 E1 FE C3 F7 F7 22 42 4F 4F 54 22 0D

130 CODE  00 00 00 00 00 00 00 00

140 REM Variable E holds the number of bytes in line 120

150 LET E=56,M=&8000,W=LCTN(120),X=LCTN(100),Y=LCTN(110),Z=LCTN(130)

160 CLS 

170 PRINT "Select option:"

180 PRINT " 0 ‑ MAKE DISK BOOTABLE"

190 PRINT " 1 ‑ REMOVE DISK BOOTABILITY"

200 PRINT " 2 ‑ EXIT"

210 PRINT 

220 REPEAT 

230   LET K=GETN

240 UNTIL K=&0030 OR K=&0031 OR K=&0032

250 IF K=&0031 THEN  GOTO 270

260 IF K=&0032 THEN  END 

270 PRINT "Input drive to patch (A‑D) : ";

280 LET S=GETN

290 PRINT CHR$(S)

300 IF S<&0041 OR S>&0044 THEN  GOTO 270

310 POKE Y,S‑&0041

320 PRINT "Insert DOS disk in drive ";CHR$(S)

330 PRINT "and press a key when ready"

340 LET Z$=GET$

350 PRINT "Reading blocks 0 and 1...."

360 PROC ACCESS(0,1,Z,S‑&0041)

370 PROC ACCESS(0,3,M,Y)

380 PROC ACCESS(1,3,M+&0100,Y)

390 IF DPEEK(M)<>1 THEN  PROC NOTDOS

400 IF K=&0031 THEN  GOTO 530

410 IF DPEEK(M+2)=0 THEN  GOTO 460

420 PRINT "Disk already contains boot code"

430 PRINT "OK to continue (Y/N) "

440 LET P$=GET$,P$=UPC$(P$)

450 IF P$="N" THEN  END 

460 PRINT "Patching in boot code...."

470 DPOKE M+2,&003B*256+&0018

480 FOR P=0 TO E‑1

490   POKE M+&003E+P,PEEK(W+P)

500 NEXT P

510 PROC WRITE

520 END 

530 PRINT "Removing boot code...."

540 DPOKE M+2,0

550 FOR P=M+&003E TO M+&0200 STEP 2

560   DPOKE P,0

570 NEXT P

580 PROC WRITE

590 END 

600 DEFPROC WRITE

610 PRINT "Writing blocks 0 and 1 to disk...."

620 PROC ACCESS(0,5,M,Y)

630 PROC ACCESS(1,5,M+&0100,Y)

640 IF K=&0030 THEN  PRINT "Disk is now BOOTABLE"

650 ELSE  PRINT "Disk is now UNBOOTABLE"

660 PRINT 

670 ENDPROC 

680 DEFPROC ACCESS(b,c,a,d)

690 POKE Y+1,b

700 DPOKE X+5,d

710 POKE X+10,c

720 CALL X,a

730 IF HL THEN  PROC DOS ERROR

740 ENDPROC 

750 DEFPROC NOTDOS

760 PRINT "Disk in drive is not a DOS disk!"

770 END 

780 DEFPROC DOS ERROR

790 PRINT "DOS error : ";#HL

800 END 

100 REM SAMPLE BOOT LOADER

110 TEXT

120 PRINT

130 PRINT "YOUR WISH IS MY COMMAND, OH MASTER"

140 PRINT

150 EXT MLOAD "B:ANOTHER.PROGRAM"

160 NEW


The sample boot loader should be saved as EXT SA."BOOT",100

Source code of the routine in line 120 of the BASIC program.

   1     ; Dos boot loader and execution 

   2     ; for the 96K and 128K Lynx 

   3     ; 

   4     ; Martyn Smith 10.04.91 

   5     ; 

   6     ;

   7     ; 

   8     ; 

   9     ; local equates 

  10     ; 

  11 DESTINATION: EQU 0FED8H ;datastore buffer (256) 

  12 STKSTOR: EQU 40E6H ;storage of stack ptr (LOWDOS) 

  13 CLOSDSK: EQU 0FC78H ;close file/disk routine 

  14 LODBAS: EQU 0F7F7H ;entry of load basic prog code 

  15 quote: EQU 22H ;ascii for speech mark 

  16 cr: EQU 0DH ;ascii for carriage return 

  17     ; 

  18     ; the loader routine must be position 

  19     ; independant code (PIC). On an EXT BOOT, 

  20     ; blocks 0 and 1 will be loaded from disk into low dos           21     ; ram at one of the circular buffers (4633H/4833H)

  22     ; Unfortunately, there is no way of telling which one,

  23     ; so the search routine will calculate 

  24     ; the location for us. 

  25     ; those of you who have disassembled the CP/M 

  26     ; system tracks may find the search routine 

  27     ; rather familiar ! 

  28     ; 

  29 SEARCH: LD HL,0 

  30     LD B,H 

  31     LD C,L 

  32     LD A,0B7H ; z80 'or a' opcode 

  33 SER10: CPIR 

  34     DEC HL 

  35     LD (HL),0 

  36 SER20: OR A 

  37     LD (HL),A 

  38     INC HL 

  39     JR NZ,SER10 

  40     ; 

  41     ; the shift routine copies the boot 

  42     ; execution code to an address in 

  43     ; user ram (from behind the roms) 

  44     ; 

  45 SHIFT: LD DE,FIN‑SER20‑1 ; offset to exec code 

  46     ADD HL,DE ; calculate physical ram location 

  47     LD DE,DESTINATION ; a safe area 

  48     LD BC,LENGTH ; length of exec code 

  49     LDIR ; copy exec code to visible ram 

  50     LD HL,(STKSTOR) ; get return address 

  51     LD DE,DESTINATION ; where exec is to cont on RET 

  52     LD (HL),E 

  53     INC HL ; overwrite old return address 

  54     LD (HL),D ; with exec addr of the boot exec 

  55     XOR A 

  56     RET ; exit from usual EXT BOOT loader 

  57     ; 

  58     ; the position of the label FIN 

  59     ; must be after the last instruction 

  60     ; of the boot loader, but before 

  61     ; the first instruction of the boot exec 

  62     ; 

  63 FIN: 

  64     ; 

  65     ORG DESTINATION ; where the exec code will be loaded 

  66     ; 

  67     ; boot execution routine 

  68     ; 

  69 BOOTC: 

  70     ; the first operation of the exec code 

  71     ; must be to close the disk. this is usually 

  72     ; done by the EXT BOOT routine BUT  

  73     ; since execution has been re-routed, it has to be 

  74     ; carried out by the boot code  

  75     ; 

  76     CALL CLOSDSK 

  77     ; 

  78     ; the next two lines are equivalent to 

  79     ; EXT LOAD "BOOT" <cr> 

  80     ; 

  81     LD DE,FILENAME 

  82     JP LODBAS ; dos takes care of the RET addr ! 

  83     ; 

  84 FILENAME: DB quote,"BOOT",quote,cr 

  85 LENGTH: EQU $‑BOOTC 

  86     END



